02-Jun-1998

Status of study of AOT02 line profiles and fluxes on NGC6543

Author: H. Feuchtgruber

0) Introduction and Purpose

The original purpose of the AOT02 observations on NGC6543 done from REV377 onwards in biweekly intervals has been primarily to monitor the SWS grating wavelength calibration on an external source, since a time dependent shift was realized.
While the wavelength calibration is not yet updated according to the results of these observations, they have been analyzed in a standardized way, in order to obtain information about SWS line profiles of unresolved lines and reproducability of line fluxes. A first report on this study is given in REF. 1.
First results on instrumental profile and resolution were also presented in REF. 2 and REF. 3.

1) Description of the available data

All processing started from SPD level with latest grating wavelength calibration (incl. time dependencies) applied. Supplementary information was extracted per observation from the SPD header and the AOCS files and stored into IDL arrays:

- Revolution number
- Rollangle
- Mean startracker y
- Mean startracker z
- Guidestar number

Also aotband, restwavelength of lines and name of the lines have been included in this set of arrays.

A total of 32 observations were carried out starting from REV377 to REV867. Every observation usually contains 13 lines, but due to time constraints in routine calibration revolutions, a few times not always all lines were observed.

2) Data processing

The observations are processed with IA by an IDL 3.6.1 script listed in appendix A, B and C. It mainly consists of:

- Read in the Files (SWSP* + AOCS*)
- Initialize supplemental arrays
- Read the SPD file
- Extract some header information
- Read the AOCS file
- Derive mean STR Y and STR Z
- Update wavelength calibration (GET_GRAT_ANG + WAVE)
- DAAR
- Flatfielding and sigma-clipping
- Derive a guess for the Gauss-fit
- Fit a gaussian to the line
- Convert to W/cm^2/um and derive lineflux
- Save lineflux and fit parameters in arrays

- Save derived AARs and store
- Loop through all lines
- Loop through all revolutions
- Store results

Due to some missing or corrupt (TM/TC, memory) lines, several checking steps in the script make sure that only useful data is selected for analysis.
In order to get clear trends for line fluxes, only one (!) CAL13 has been used to process all the observations (CAL13_017). Otherwise, all calfiles and the TIMEDEP file from OLP 7.0 are used.

3) Results

3.1 Line fluxes

The line fluxes of every line and every observation is plotted against revolution in figure 1 (flux units: [1e-18 W/cm^2]). For almost all lines a modulation of the line flux of the order of 10% can be seen. This modulation can be attributed to the spatial extension of NGC6543 versus the SWS slit size. The slight elliptical shape (at VIS, IR and Radio wavelengths) and the rectangular slit easily produce a modulation as the one observed when the rollangle varies along the year (see figure 5, CAM-CVF at 8.943 um; Reay and Worswick, MNRAS,199,1982 for radio image; AJ, 94,1987, visible light plates; SWS slit vs. radio image see REF. 1)
Some scatter in the measured line fluxes might come from the use of different guide stars at different angular distance from NGC6543, which can cause slight pointing errors due to the different aberration between target and guidestar. Also, there might be still some residual pointing errors when guidestars are close to edge of the startracker, caused by inaccuracies in the startracker focal length correction. Figure 4 gives the position of the guidestar for every observation together with guidestar number and rollangle.
Observations by PHOT-S gave a larger line flux for the ArIII line at 8.99 micron, but the aperture was also larger there.

3.2 Line profiles

While all the individual line data is stored in the form of AARs, the results of fitting a gaussian give FWHM or resolution values for every line. There is still a lack of definition how SWS line profiles should be stored (rebinned ?, raw ?, per detector ?, etc.; see also discussion at REF.1.). The profiles of the H-recombination lines show a blue wing, most likely caused by some He lines also present on NGC6543 (see also figure 6), therefore these lines will not be suitable for accurate profile determination. The ArIII line at 21.8 micron suffers from fringing in the continuum, also this line is not suitable for profile determination down to the few percent level. The strong lines mainly in band 4 show excessive noise at the peaks (see figure 6) which might be due to detector hysteresis effects, affecting the fit and also the line profile to some extent.
The resolution (=lambda/(FWHM of fitted gaussian)) of every line and every observation is plotted in figure 2. The resolution remains almost constant in time, variations are of the order of a few percent and could also be caused by the geometry orientation of the target vs. the dispersion direction of SWS. The actual resolution values given are in general somewhat lower than the ideal point source resolutions given by the IA module "RESOLUTION", only the 33.48 and 36 micron line appear to have some higher resolution than theoretically predicted.

3.3 Line centers

One more result of the gaussian fit to the lines is the line center. Figure 3 shows a plot of the line centers corrected for heliocentric velocity of NGC6543 (=-66.1 km/s) and ISO's velocity towards target. Restwavelengths of the individual lines are given as subtitles. The TIMEDEP switches from CAL16E_013 to CAL16E_014 at REV420 and from CAL16E_014 to CAL16E_015 at REV581. These steps are clearly seen in the trend of the line centers. The accuracy of expected line center vs. measured line center is even with the apparent trends still better than 1e-4 and can easily be improved when a more detailed time dependent wavelength calibration is derived.

4) Open Problems

All results are stored as IDL save sets. As soon as a clear definition of the type and shape of a "Instrumental Profile Calfile" is defined these observations can contribute to its contents.

5) Further documents

REF. 1: SWS Ground Segment Meeting 12-13 March 1998, Minutes
REF. 2: The wavelength calibration and resolution of SWS,
A&A, 315, L60, 1996
REF. 3: Report on SWS Web on instrumental profile from M. v. d. Ancker

6) Figure captions

Figure 1:
Lineflux in [1e-18 x W/cm^2 vs. revolution number; name of line and restwavelength are given as well

Figure 2:
Resolution (in lambda/(FWHM gaussian fit)) vs. revolution number; name of line and restwavelength are given as well

Figure 3:
Observed (and velocity corrected) wavelength vs. revolution number for every line; The horizontal line indicates the restwavelength;

Figure 4:
Guidestar position on startracker field of view; the different symbols indicate different guidestars, the legend gives the guidestar numbers; The upper number close to each symbol gives the revolution number, the lower number represents the rollangle

Figure 5:
CAM-CVF image of NGC6543 at 8.943 micron

Figure 6:
Br-Alpha and [SIII] 33.48 micron line on NGC6543 as typical examples of profiles obtained within the AOT02s on this target.

Appendix A

Listing of script to analyse the NGC6543 AOT02 observations:

```
==============================================================
; initialize
path='dkb100:[fgb.data.ngc6543]'
file=['SWSP37700438','SWSP39101474','SWSP40600401','SWSP41901703',$
      'SWSP44003903', $
```

```
       'SWSP46800602',$
       'SWSP48201102',$
       'SWSP51700604',$
       'SWSP53103502','SWSP54400402',$
       'SWSP55206604',$
       'SWSP56602405',$
       'SWSP58700902','SWSP60101202','SWSP61503002','SWSP62902102',$
       'SWSP64300801','SWSP65700601','SWSP67101301','SWSP68501401',$
       'SWSP69900801','SWSP71300301','SWSP72700301','SWSP74101301',$
       'SWSP75500401','SWSP76902301','SWSP78300301','SWSP79700201',$
       'SWSP82701001','SWSP83900401','SWSP85300301','SWSP86700201']

lines=[2.62587,3.29700,4.05226,4.65378,7.45986,8.99138,10.5105,12.81355,15.5551
       18.7130,21.8293,33.4810,36.0135]
aotband=['1b','1d','1e','2a','2c','2c','2c','3a','3a','3c','3d','4','4']
name=['H64','H95','H54','H75','H65','ArIII8','SIV','NeII','NeIII15','SIII18',$
       'ArIII21','SIII33','NeIII36']
lineflux=fltarr(n_elements(file),n_elements(lines))
result=fltarr(n_elements(file),n_elements(lines),4)
roll=fltarr(n_elements(file))
rev=fltarr(n_elements(file))
stry=fltarr(n_elements(file))
strz=fltarr(n_elements(file))
guidestar=fltarr(n_elements(file))
restore,'dkb100:[fgb.data.ngc6543]colguideforngc6543.idl'
; conversion to arcsec
sc=180.*60*60/!pi
; do it now for all revolutions available
for i=0,n_elements(file)-1 do begin $
  spd=read_fspd(path+file(i)+'.fits')
  a=read_fits_key(spd.header,'INSTROLL',r,comment)
  roll(i)=float(r)
  a=read_fits_key(spd.header,'FILENAME',r,comment)
  rev(i)=float(strmid(r,4,3))
  a=read_fits_key(spd.header,'ATTGUIDE',r,comment)
  guidestar(i)=float(r)
  aocs=read_faocs(path+'aocs'+strmid(file(i),4,8)+'.fits')
  stry(i)=mean(float(aocs.data.stry/839825.)*sc)
  strz(i)=mean(float(aocs.data.strz/839825.)*sc)
print,rev(i),roll(i),guidestar(i)
  spd=get_grat_ang(spd)
  spd=wave(spd); ,cal16a='ias_root:[cal]cal16a_010.fits')
  aar=daar(spd,cal13='ias_root:[cal]cal13_017.fits')
; do it for all lines in that revolution
  for j=0,n_elements(lines)-1 do begin $
    upper=lines(j)+lines(j)*10./2000.
    lower=lines(j)-lines(j)*10./2000.
    ind=where(aar.data.wave ge lower and aar.data.wave le upper)
; do it only if data points exist for that line, because not all lines were
; done in all revolutions
    if n_elements(ind) ge 10 then begin $
; cut out your line and do some standard processing
    cutaar,aar,test,lower,upper,aotband(j)
    test=sws_flatfield(test,lines=0,/noplot,aot_band=aotband(j))
    test=sigclip(test,sig=3,lines=0,nit=4,res=1800,aot_band=aotband(j),/noplot)
; get initial estimates for your fit to the data
    dummy=sws_rebin(test,resol=5000,over=6,lines=0,method='mean',/noplot)
    guess=gauss_fit(dummy.data.wave,dummy.data.flux,a)
    stat=a
; this takes the guess and does the actual fitting
    g=gauss_fit_ngc6543(test.data.wave,test.data.flux,stat)
```

```
; now do all kind of conversions for various types of investigations
    fit=test
    fit.data.flux=g
    ratio=fit
    ratio.data.flux=ratio.data.flux/test.data.flux
    dummy2=spect_form(dummy,'w/cm^2/um')
    dummy2.data.flux=dummy2.data.flux*1.0e18
    yfit=gauss_fit(dummy2.data.wave,dummy2.data.flux,a)
; derive the lineflux from the fit to the converted data and store all
; results of interest in the variables lineflux and result
; i = revolutions, j = lines, * = the four fitting parameters
    lineflux(i,j) = a(0)*sqrt(2*!pi* a(2)^2)*1.0e-18
    result(i,j,*)=stat
; do some output control
  print,lineflux(i,j)
    plotaar,test
    plotaar,fit,/oplot,psym=-3
    plotaar,dummy,/oplot,psym=-3
; save the individual data of a line
save,file='dkb100:[fgb.data.ngc6543]aot02_'+strtrim(string(fix(rev(i))),2)+$
          '_'+strtrim(string(j),2)+'.xdr',test,dummy,fit,ratio,/xdr
    endif
  endfor
endfor
; save your results in a file
save,file='dkb100:[fgb.data.ngc6543]allrollangles.idl',rev,roll,stry,strz,$
                                                        lines,name,guidestar
save,file='dkb100:[fgb.data.ngc6543]resultsfor_17.idl',lineflux,result
end
==========================================================
```

Appendix B

Code for CUTAAR, called by main script:

```
==========================================================
pro cutaar,inst,outst,lmin,lmax,aotband

if aotband ne '' then begin $
aotband=strupcase(aotband)

band=['1A','1B','1D','1E','2A','2B','2C','3A','3C','3D','3E','4','5A','5B', $
      '5C','5D','6']
ord =[4,3,3,2,2,1,1,2,2,1,1,1,3,2,2,1,1]
slit=[1,1,2,2,2,2,3,1,2,2,3,3,1,1,2,2,3]
deth=[12,12,12,12,24,24,24,36,36,36,36,48,49,49,49,49,51]
detl=[1,1,1,1,13,13,13,25,25,25,25,37,49,49,49,49,51]

ind=where(band(*) eq aotband)
dummy1=ord(ind)
dummy2=slit(ind)
dummy3=detl(ind)
dummy4=deth(ind)

index=where(inst.data.wave ge lmin and $
            inst.data.wave le lmax and $
            byte(inst.data.flag) eq (dummy1(0)*32) and $
            test_status(inst.data.status,aper=dummy2(0)) eq 1 and $
            inst.data.det le dummy4(0) and inst.data.det ge dummy3(0))
```

```
endif else begin $

index=where(inst.data.wave ge lmin and $
            inst.data.wave le lmax and $
            (byte(inst.data.flag) and 1) ne 1 and $
            (byte(inst.data.flag) and 2) ne 2 and $
            (byte(inst.data.flag) and 16) ne 16)
endelse
if n_elements(index) gt 1 then $
 outst=define_aar(header=inst.header,length=n_elements(index),$
                  data=inst.data(index)) else outst=0
end
===================================================================
```

Appendix C

Code for GAUSS_FIT_NGC6543, called by main script:

```
pro     gaussian,x,a,f,pder
z = (x-a(1))/a(2)                    ;get z
ez = exp(-z^2/2.)*(abs(z) le 7.)     ;gaussian part ignore small terms
f = a(3)+a(0)*ez                     ;function.
if n_params(0) le 3 then return      ;need partial?
pder = fltarr(n_elements(x),4)       ;yes, make array.
pder(0,0) = ez                       ;compute partials...
pder(0,1) = a(0) * ez * z/a(2)
pder(0,2) = pder(*,1) * z
pder(*,3) = 1.
return
end
function gauss_fit_ngc6543,x,y,a
;+
; NAME:
;       GAUSS_FIT
; PURPOSE:
;       fit y=f(x) where:
;       f(x) = a0*exp(-z^2/2) + a3
;               and z=(x-a1)/a2
;       a0 = height of gaussian, a1 = center of gaussian, a2 = 1/e width,
;       a3 = background.
;       Estimate the parameters a0,a1,a2,a3 and then call curfit.
; CATEGORY:
;       ?? - fitting
; CALLING SEQUENCE:
;       yfit = gauss_fit(x,y,a)
; INPUTS:
;       x = independent variable, must be a vector.
;       y = dependent variable, must have the same number of points
;               as x.
;       quiet = set to inhibit printing curfit iterations.
; OUTPUTS:
;       yfit = fitted function.
; OPTIONAL OUTPUT PARAMETERS:
;       a = coefficients. a three element vector as described above.
;
; COMMON BLOCKS:
;       None.
; SIDE EFFECTS:
;       None.
; RESTRICTIONS:
```

```
; PROCEDURE:
; MODIFICATION HISTORY:
;       Adapted from GAUSSFIT
;       D. L. Windt, AT&T Bell Laboratories, March, 1990
;       Adapted by HF, to force use of input parameter estimates
;-
;
on_error,2
cm=check_math(0.,1.)            ; Don't print math error messages.
n = n_elements(y)               ; # of points.
c=poly_fit(x,y,1,yf)            ; Do a straight line fit.
yd=y-yf
ymax=max(yd) & xmax=x(!c) & imax=!c     ;x,y and subscript of extrema
ymin=min(yd) & xmin=x(!c) & imin=!c

if abs(ymax) gt abs(ymin) then i0=imax else i0=imin ;emiss or absorp?
i0 = i0 > 1 < (n-2)             ;never take edges
dy=yd(i0)                       ;diff between extreme and mean
del = dy/exp(1.)                ;1/e value
i=0
while ((i0+i+1) lt n) and $     ;guess at 1/2 width.
        ((i0-i) gt 0) and $
        (abs(yd(i0+i)) gt abs(del)) and $
        (abs(yd(i0-i)) gt abs(del)) do i=i+1
; HF: Now do not change a and comment it out
;a = [yd(i0), x(i0), abs(x(i0)-x(i0+i)),c(0)] ;estimates

!c=0                            ;reset cursor for plotting
return,curfit(x,y,replicate(1.,n),a,sigmaa,funct='gaussian',/quiet) ;call curfit
end
==============================================================
```


H64
2.62587 micron
H95
3.29700 micron
H54
4.05226 micron
H75
4.65378 micron
H65
7.45986 micron
ArIII8
8.99138 micron
SIV
10.5105 micron
NeII
12.8135 micron
NeIII15
15.5551 micron
SIII18
18.7130 micron
ArIII21
21.8293 micron
SIII33
33.4810 micron
NeIII36
36.0135 micron


H64
Restwavelength: 2.62587 micron
H95
Restwavelength: 3.29700 micron
H54
Restwavelength: 4.05226 micron
H75
Restwavelength: 4.65378 micron
H65
Restwavelength: 7.45986 micron
ArIII8
Restwavelength: 8.99138 micron
SIV
Restwavelength: 10.5105 micron
NeII
Restwavelength: 12.8135 micron
NeIII15
Restwavelength: 15.5551 micron
SIII18
Restwavelength: 18.7130 micron
ArIII21
Restwavelength: 21.8293 micron
SIII33
Restwavelength: 33.4810 micron
NeIII36
Restwavelength: 36.0135 micron


H64
Restwavelength: 2.62587 micron
H95
Restwavelength: 3.29700 micron
H54
Restwavelength: 4.05226 micron
H75
Restwavelength: 4.65378 micron
H65
Restwavelength: 7.45986 micron
ArIII8
Restwavelength: 8.99138 micron
SIV
Restwavelength: 10.5105 micron
NeII
Restwavelength: 12.8135 micron
NeIII15
Restwavelength: 15.5551 micron
SIII18
Restwavelength: 18.7130 micron
ArIII21
Restwavelength: 21.8293 micron
SIII33
Restwavelength: 33.4810 micron
NeIII36
Restwavelength: 36.0135 micron

Guidestar Positions on STR for NGC6543 AOT02s






NGC6543 H 5−4
Flux [Jy]
Wavelength [μm]
NGC6543 SIII
Flux [Jy]
Wavelength [μm]